

NAG Library Routine Document

G05SFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05SFF generates a vector of pseudorandom numbers from a (negative) exponential distribution with mean a .

2 Specification

```
SUBROUTINE G05SFF(N, A, STATE, X, IFAIL)
  INTEGER          N, STATE(lstate), IFAIL
  double precision A, X(N)
```

3 Description

The exponential distribution has PDF (probability density function):

$$f(x) = \frac{1}{a}e^{-x/a} \quad \text{if } x \geq 0,$$

$$f(x) = 0 \quad \text{otherwise.}$$

G05SFF returns the values

$$x_i = -a \ln y_i$$

where y_i are the next n numbers generated by a uniform $(0, 1]$ generator.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SFF.

4 References

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin
 Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

5 Parameters

- 1: N – INTEGER *Input*
On entry: n , the number of pseudorandom numbers to be generated.
Constraint: $N \geq 0$.
- 2: A – **double precision** *Input*
On entry: a , the mean of the distribution.
Constraint: $A > 0.0$.
- 3: STATE(*lstate*) – INTEGER array *Communication Array*
Note: *lstate* = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.
On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

- 4: $X(N)$ – *double precision* array *Output*

On exit: the n pseudorandom numbers from the specified exponential distribution.

- 5: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 0$.

IFAIL = 2

On entry, $A \leq 0.0$.

IFAIL = 3

On entry, STATE vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints five pseudorandom numbers from an exponential distribution with mean 1.0, generated by a single call to G05SFF, after initialization by G05KFF.

9.1 Program Text

```
*      G05SFF Example Program Text
*      Mark 22 Release. NAG Copyright 2006.
*      .. Parameters ..
      INTEGER                NOUT
      PARAMETER              (NOUT=6)
      INTEGER                MSTATE, MSEED, N
      PARAMETER              (MSTATE=633,MSEED=1,N=5)
*      .. Local Scalars ..
      DOUBLE PRECISION       A
      INTEGER                GENID, I, IFAIL, LSEED, LSTATE, SUBID
*      .. Local Arrays ..
```

```

      DOUBLE PRECISION X(N)
      INTEGER          SEED(MSEED), STATE(MSTATE)
*    .. External Subroutines ..
      EXTERNAL          G05KFF, G05SFF
*    .. Executable Statements ..
      WRITE (NOUT,*) 'G05SFF Example Program Results'
      WRITE (NOUT,*)
*    Set the distribution parameter A
      A = 1.0D0
*    Initialize the seed to a repeatable sequence
      SEED(1) = 1762543
*    GENID and SUBID identify the base generator.
      GENID = 1
      SUBID = 1
*    Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 20
      END IF
*    Generate the variates
      IFAIL = 1
      CALL G05SFF(N,A,STATE,X,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 20
      END IF
*    Display the variates
      WRITE (NOUT,99999) (X(I),I=1,N)
*
      20 CONTINUE
*
99999 FORMAT (1X,F10.4)
99998 FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997 FORMAT (1X,' ** G05SFF returned with IFAIL = ',I5)
      END

```

9.2 Program Data

None.

9.3 Program Results

G05SFF Example Program Results

```

0.4520
2.2398
0.2930
0.2253
2.2577

```
